 

<retazec> konstanta ale moze byt aj vyraz
napr
 

"ABC"
"A" + B
 

Left( 2, "A" + Trim(B))
(B je premenna)
 

<string> je retazec, pri spracovavani ktoreho je dolezita velkost pismen
<text> je retazec, pri spracovavani ktoreho nie je dolezita velkost pismen
 

 

 

Funkcie - t.j. ze ich mozno pouzit vo vyraze
====================================

Line( <riadok 1..$FFFF>, <retazec> )
Vrati z retazca urceny riadok

LineWithText( <text>, <retazec> )
Vrati z retazca taky prvy riadok, kde sa nachadza zadany text (text znaci ze 
nezalezi na velkosti pismen)

LineWithString( <string>, <retazec> )
Vrati z retazca taky prvy riadok, kde sa nachadza zadany string (string 
znaci ze zalezi na velkosti pismen)

Left( <zobrat pocet od zaciatku>, <retazec> )
Vrati z retazca zadany pocet znakov zlava (alebo menej, ak tam tolko nie je)
V ByLines rezime to urobi pre kazdy riadok

Right( <zobrat pocet od konca>, <retazec> )
Vrati z retazca zadany pocet znakov sprava (alebo menej, ak tam tolko nie 
je)
V ByLines rezime to urobi pre kazdy riadok

Mid( <preskocit>, <zobrat pocet>, <retazec> )
Preskoci zadany pocet znakov (zlava) a vrati nasledujucich <pocet> znakov z 
retazca (alebo menej, ak tam tolko nie je)
V ByLines rezime to urobi pre kazdy riadok

Trim( <retazec> )
Vrati retazec, z ktoreho oreze biele znaky zo zaciatku aj z konca
V ByLines rezime to urobi pre kazdy riadok

TrimLeft( <retazec> )
Vrati retazec, z ktoreho oreze biele znaky zo zaciatku
V ByLines rezime to urobi pre kazdy riadok

TrimRight( <retazec> )
Vrati retazec, z ktoreho oreze biele znaky z konca
V ByLines rezime to urobi pre kazdy riadok

TrimLeftToText( <text>, <retazec> )
Vrati retazec, z ktoreho oreze znaky zo zaciatku, tak aby retazec ktory 
vrati zacinal zadanym textom (nezalezi na velkosti pismen)
V ByLines rezime to urobi pre kazdy riadok

TrimLeftToString( <string>, <retazec> )
Vrati retazec, z ktoreho oreze znaky zo zaciatku, tak aby retazec ktory 
vrati zacinal zadanym stringom (zalezi na velkosti pismen)
V ByLines rezime to urobi pre kazdy riadok

IntFromLeft( <poradove cislo cisla 1..N od zaciatku>, <oddelovace>, <retazec> )
Vrati cele cislo zlava )od zaciatku), za cislo sa povazuje zhluk cifier, 
pred ktorym moze byt znamienko a okolo musi byt niektory z oddelovacov, 
poradie cisla sa urci podla poctu cisel, nie podla poctu oddelovacov
V ByLines rezime to urobi pre kazdy riadok

IntFromRight( <poradove cislo cisla 1..N od konca>, <oddelovace>, <retazec> )
Vrati cele cislo sprava (od konca), za cislo sa povazuje zhluk cifier, pred 
ktorym moze byt znamienko a okolo musi byt niektory z oddelovacov, poradie 
cisla sa urci podla poctu cisel, nie podla poctu oddelovacov
V ByLines rezime to urobi pre kazdy riadok

NumFromLeft( <poradove cislo cisla 1..N od zaciatku>, <desatinne oddelovace>, 
<oddelovace>, <retazec> )
Vrati cislo zlava (od zaciatku), za cislo sa povazuje zhluk cifier, pred 
ktorym moze byt znamienko a okolo musi byt niektory z oddelovacov, a moze 
tam byt max jeden desatinny oddelovac, poradie cisla sa urci podla poctu 
cisel, nie podla poctu oddelovacov
V ByLines rezime to urobi pre kazdy riadok

NumFromRight( <poradove cislo cisla 1..N od konca>, <desatinne oddelovace>, 
<oddelovace>, <retazec> )
Vrati cislo sprava (od konca), za cislo sa povazuje zhluk cifier, pred 
ktorym moze byt znamienko a okolo musi byt niektory z oddelovacov, a moze 
tam byt max jeden desatinny oddelovac, poradie cisla sa urci podla poctu 
cisel, nie podla poctu oddelovacov
V ByLines rezime to urobi pre kazdy riadok

EncodeBase64( <retazec> )
Vrati retazec zakodovany v BASE64

DecodeBase64( <retazec> )
Vrati retazec dekodovany z BASE64

ByLines( <retazec> )
Zapina spracovanie retazca po riadkoch. Ma vyznam, ak retazec je vyraz, t.j.
Ak

SET A "123" + CRLF + "456"
SET B "ABC" + CRLF + "DEF"

Tak

SET C A+B
sposobi ze C = "123" + CRLF + "456ABC" + CRLF + "DEF"

SET D ByLines( A+B )
sposobi ze D = "123ABC" + CRLF + "456DEF" + CRLF


ByString( <retazec> )
Opacna funkcia ako ByLines, ucitocna pre pripady, ze v ramci spracovania 
treba nieco spojit, spracovat nie po riadkoch, ale ako text, t.j.

SET D ByLines( ByString( A+B ))  je to iste ako SET D A+B


SubstByLines( <substitucia>, <retazec> )
Vrati retazec, v ktorom kazdy riadok nahradi prislusnou substituciou (ak 
taku najde). Na velkosti pismen zalezi.
Substitucia je retazec, kde kazdy riadok predstavuje jednu podmienku, a to v tvare <substituovany text>=<novy text>
Napr.

SET S "0=OFF" + CRLF + "1=ON"
SET R "1" + CRLF  + "1" + CRLF + "0" + CRLF + "0"

SET X SubstByLines( S, R )

Sposobi ze premenna X bude "ON" + CRLF + "ON" + CRLF + "OFF" + CRLF + "OFF"
====================================

v dalsom texte vyraz, decsep su <retazec> (t.j. ze musia byt v uvodzovkach)
decsep obsahuje znaky ktore su povazovane za desatinny oddelovac - t.j. napr. ",."
decimals je cislo, urcuje pocet desatinnych miest
 

Set( varname, vyraz )
kde ked sa zapise do varname @a, tak sa nebude priradovat priamo do premennej a, ale do premennej ktorej meno je v premennej a

tato funkcia aj nastavi hodnotu do premennej, aj ju vrati.

 

MulDiv( decimals, decsep, mul, div, vyraz )
vyraz vynasobi mul (celociselna hodnota) a vydeli div (celociselna hodnota), pricom vyraz (po vyhodnoteni) musi byt cislo, biele znaky sa orezu, funguje aj bylines mod, ako desatinny oddelovac je mozne pouzit jeden zo znakov nastavenych v decsep, a zaokruhli sa to na decimals desatinnych miest (aj sa sformatuje, desatinny oddelovac je .)

 

napr MulDiv( 2, ",.", 1, 2, "100") vrati "50.00"

 

 

Mul( decimals, decsep, vyraz1, vyraz2 )
vrati vysledok nasobenia vyraz1 x vyraz2, zaokruhleny na decimals miest (moze byt aj na 0), funguje aj v bylines mode, (ked je je v oboch vektoroch viac ako 1 prvok, tak sa v tom mensom doplnia 0)

 

Cize ByLines( Mul (2,",.","10","1"+CR+"2"+CR+"3")) vyrobi

 

10.00

20.00

30.00

 

ale 

 

Cize ByLines( Mul (2,",.","10"+CR+"20","1"+CR+"2"+CR+"3")) vyrobi

 

10.00

40.00

0.00

 

Div( decimals, decsep, vyraz1, vyraz2 )
vrati vysledok delenia vyraz1 / vyraz2, zaokruhleny na decimals miest (moze byt aj na 0), funguje aj v bylines mode, (ked je je v oboch vektoroch viac ako 1 prvok, tak sa v tom mensom doplnia 0) => hrozi delenie 0

 

Add( decimals, decsep, vyraz1, vyraz2 )
vrati vysledok scitania vyraz1 + vyraz2, zaokruhleny na decimals miest (moze byt aj na 0), funguje aj v bylines mode, (ked je je v oboch vektoroch viac ako 1 prvok, tak sa v tom mensom doplnia 0)

 

Sub( decimals, decsep, vyraz1, vyraz2 )
vrati vysledok odcitania vyraz1 - vyraz2, zaokruhleny na decimals miest (moze byt aj na 0), funguje aj v bylines mode, (ked je je v oboch vektoroch viac ako 1 prvok, tak sa v tom mensom doplnia 0) 

 

 

 

Match( mode, pattern, string )
funguje aj v bylines rezime

 

mode je rovnaky vyznam ako ma premenna MatchMode, dokonca mozno priamo zapisat

 

Match( matchmode,...

 

pattern je to co sa porovnava (cize napr. ten regularnyvyraz)

string je s cim sa porovnava (moze byt aj premenna)

 

v pripade ze to matchuje sa vrati cely string, v pripade ze nie, tak nic.

v bylines rezime, vrati vysledok, ktory obsahuje tie riadky, ktore matchuju. Ak ziaden, tak vrati nic (prazdny retazec).

 

ak sa pouzivaju regularne vyrazy (dane parametrom mode), tak dochadza k naplneniu aj premennej 

 

matchsubexprN

 

ktora v bylines rezime obsahuje matchujuce casti riadku (ak cely riadok matchuje)

 

cize priklad ktory je nizsie robi to, co by mal, a co bolo pozadovane

 

Ak su potrebne premenne matchsubexpr, odporucam nevnarat match(..match(.. 

pretoze k vyhodnocovaniu dochadza tak, ze sa najskor vyhodnoti to co je najvnorenejsie, a naposledy to co je najmenej vnorene a na konci (parametre sa vyhodnocuju od zaciatku)

 

 

=======================================================

 

SET A " eth0 9908 168135"+CR+" eth2 0 0"+CR+" eth1 0 0"+CR+" sit0 0 0"+CR+" ham0 0 0"+CR+CR+"sealit@fileserver:~$"+CR+"sealit@fileserver:~$"
SET A ByLines(trim(A))

 

SHOW A
show "================"

 


set b ByLines(match( "regexpr", "([a-z0-9]+) ([0-9]+) ([0-9]+)", A ))
show b
show "================"

 

show ByLines(matchsubexpr1+" "+matchsubexpr2+" "+matchsubexpr3)
show "================"


 

 

 

 

If( <vyraz>, <retazec_true>, <retazec_false> )
ak je vyraz neprazdny, tak vrati <retazec_true>, ak je prazdny, vrati <retazec_false>
V ByLines rezime to urobi pre kazdy riadok

 

And( <operand>[, <operand>,...] )
logicky sucin, pricom za FALSE sa povazuje prazdny retazec, a TRUE neprazdny. Ako logicku jednotku vrati T, ako nulu vrati prazdny retazec. Pocet operandov je 1..100.

V ByLines rezime to urobi pre kazdy riadok

 

Or( <operand>[, <operand>,...] )
NAnd( <operand>[, <operand>,...] )
NOr( <operand>[, <operand>,...] )
 

podobne ako AND

 

Not( <vyraz> )
Logicka negacia, ak je vyraz prazdny retazec, vrati "T", ak neprazdny, vrati ""

V ByLines rezime to urobi pre kazdy riadok

 

CmpStr( <retazec1>, <operator>, <retazec2> )
Porovna retazce retazec1 a retazec2 ako retazce, t.j. zalezi na velkosti pismen

Ak je podmienka splnena, vrati "T", ak nie, vrati ""

Operator je </<=/=/>=/>/<>

 

Napr. 

 

CmpStr( "BBB",">","AAA") vrati "T", lebo je pravda "BBB" > "AAA"

CmpStr( "BBB","<","AAA") vrati "", lebo nie je pravda "BBB" > "AAA"

 

V ByLines rezime to urobi pre kazdy riadok

 

CmpText( <text1>, <operator>, <text2> )
Porovna retazce text1 a text2 ako texty, t.j. nezalezi na velkosti pismen

Ostatne ako CmpStr

 

CmpNum( <retazec1>, <operator>, decsep, <retazec2> )
Porovna retazce retazec1 a retazec2 ako cisla

 

(t.j. "2" > "11" ked sa porovnava ako text, ale "2" < "11" ked sa porovnava ako cislo)

 

 

Lines( <line>, <line>...)
vyrobi vektor (viacriadkovy text). Pocet parametrov 1..100

 

Lines("aa","bb","cc") je ekvivalentne  "aa" + CRLF + "bb"  + CRLF + "cc"  + CRLF

 

 

RemoveLines( vyraz, premenna1, premenna2,...)
vymaze zvolene riadky z uvedenych premennych. Pocet premennych je max 100

Vymaze z premennych take riadky, v ktorych <vyraz> je prazdny.

<vyraz> je vyhodnocovany ako bylines (t.j. aj ked sa nezapne bylines rezim)

 

Vystup tejto funkcie je retazec, ktory obsahuje iba tie riadky, ktore zostali, a nakazdom riadku je hodnota vyrazu.

Nazornejsie to bude vidiet pri spusteni prikladu 

 

 

==========================================

Priklad

==========================================

 

 

set A "11"+CR+"22"+CR+"33"+CR+"44"
set A lines("1","2","3","4","5","6",A)

 

show A
show
show "cmpnum"
show ByLines(if(cmpnum(A,">","2"),A,A+A+A))

 

show
show "cmptext"
show ByLines(if(cmptext(A,">","2"),A,A+A+A))
show

 


show
show "cmpstr"
show ByLines(if(cmpstr(A,">","2"),A,A+A+A))
show

 

set X Lines("","1","","1","","1","","1")
set Y Lines("","","1","1","","","1","1")
set Z Lines("","","","","1","1","1","1")

 

show "AND"
show ByLines(if( And( X,Y,Z),"XXX","---"))
show
show "OR"
show ByLines(if( Or( X,Y,Z),"XXX","---"))
show

 

show "NOT AND"
show ByLines(if( Not(And( X,Y,Z)),"XXX","---"))
show
show "NOT OR"
show ByLines(if( Not(Or( X,Y,Z)),"XXX","---"))
show

 

show "NAND"
show ByLines(if( NAnd( X,Y,Z),"XXX","---"))
show
show "NOR"
show ByLines(if( NOr( X,Y,Z),"XXX","---"))
show

 


show "AND iba X"
show ByLines(if( And( X),"XXX","---"))
show
show "OR iba X"
show ByLines(if( Or( X),"XXX","---"))
show

 


show 
show "removelines"
set M lines("1","2","","3","4")
set N lines("11","22","33","44","55")
set P lines("19","29","39","49","59")

 

show M
show
show N
show
show P

 

show "==============================="
show removelines(M,M,N,P)
show M
show
show N
show
show P

 

show "==============================="
show removelines(CmpNum(M,">","2"),M,N,P)

 

show M
show
show N
show
show P

 

 

SortLines( mode, decimalsep, kluc, premenna1, premenna2,... premenna100)
usporiada riadky zvolenym sposobom

mode urcuje ako sa ma usporiadat (smer, velkost pismen..), a musi byt jedna z hodnot

ascstr  - vzostupne retazce, zalezi na velkosti pismen
dscstr - zostupne
asctext - vzostupne retazce, NEzalezi na velkosti pismen
dsctext
ascnum - vzostupne cisla, vtedy sa pouzije aj decimalsep. Porovnava sa 
automaticky bud vsetky ako integery, alebo ako extended hodnoty, zavisi, ci 
sa v niektorej nachadza uvedeny decimalseparator.
dscnum
 

kluc - retazec (t.j. bud primao zadany retazec, vyraz, ci premenna...) ktorej riadky budu usporiadane zvolenym sposobom - kedze je to vyraz, hodnota sa nezmeni. Aby sa usporiadali riadky v nejakej premennej, tak ju treba dat aj do kluca (tam sa zisti, ako sa maju riadky poprehadzovat), a aj do premennej, v ktorej sa maju takto poprehadzovat.

 

premenna1..100 - meno premennej, ktorej riadky budu preskupene podla usporiadania kluca

 

cize ked sa 

 

Priklad:

set M lines("A","B","C","D","E")
set N lines("11","2.345","3","1","5")
set P lines("19","29","39","49","59")


show bylines(M + ":" + N + ":"+ P )
show

show sortlines( "ascnum", ",.", N, M, N, P )
show
show bylines(M + ":" + N + ":"+ P )
show

show sortlines( "ascstr", "", N, M, N, P )
show
show bylines(M + ":" + N + ":"+ P )
show

=========================================

GetColumns( strings, delimiters, columns, varname1, varname2..varnameN )
 

columns je treba zadat v tvare "1,2,11,23"  oddelovat sa moze ciarkou alebo bodkociarkou, a musi to byt v uvodzovkch (je to parameter), a za tym musi ist tolko nazvov premennych , kolko je stlpcov. Viac bude zrejme z dema.

Delimiters - retazec max 10 znakov

 

parsuje sa to tak, ze sa zacina od zaciatku, a kym to nie je delimiter, tak je to sucast prveho stlpca.

potom kym je znak delmiter, je to medzera

potom kym znak nie je delimiter, je to sucast druheho stlpca.

 

atd.

 

Ze je znak delimiter znamena, ze sa nachadza v retazci delimiters, t.j. ak zadas delimiters

",;: " tak delimiter je , aj ; aj : aj medzera + vsetko mensie ako medzera. Medzera znamena defacto biely znak #0.." ".

 

Otestovat to mozes tak, ze si to v konzole spustis (co je prilozene), je to myslim dost nazorne.

 

===========================

Priklad

===========================

 

set a1 "1 lo softwareLoopback 16436 10000000 0 up up 0:0:00:00.00 523773296 4089011 0 0 0 0 523773296 4089011 0 0 0 0 SNMPv2-SMI::zeroDotZero"
set a2 "2 eth0 ethernetCsmacd 1500 10000000 0:e0:81:11:17:54 up up 0:0:00:00.00 2603125831 193641758 0 0 0 0 2720228299 109063718 0 0 0 0 SNMPv2-SMI::zeroDotZero"
set a3 "3 vmnet1 ethernetCsmacd 1500 10000000 0:50:56:c0:0:1 up up 0:0:00:00.00 0 911143 0 0 0 0 0 697699 0 0 0 0 SNMPv2-SMI::zeroDotZero"
set a4 "4 ham0 ethernetCsmacd 1200 10000000 0:ff:a8:f3:21:8c up up 0:0:00:00.00 2704446093 4610754 0 0 0 0 3848488457 4994643 0 0 0 0 SNMPv2-SMI::zeroDotZero"
set a5 "7 ham1 ethernetCsmacd 1200 10000000 0:ff:7c:6e:82:a7 up up 0:0:00:00.00 0 0 0 0 0 0 0 0 0 0 0 0 SNMPv2-SMI::zeroDotZero"

 

set a lines( a1, a2, a3, a4, a4 )

 

show a

 

show ByLines(GetColumns( TrimLeft( a ), " ", "1,2,8,10,16", index, meno, status,  send, receive ))

 

show  "=========="
show  index
show  "----------"
show  meno
show  "----------"
show  status
show  "----------"
show  send
show  "----------"
show  receive
show  "=========="

 

show GetColumns( TrimLeft( "1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17" ), " ", "1,2,8,10,16", index, meno, status,  send, receive )

 

show  "=========="
show  index
show  "----------"
show  meno
show  "----------"
show  status
show  "----------"
show  send
show  "----------"
show  receive
show  "=========="

 

 

 


Prikazy
====================================

IF - ELSEIF - ELSE - ENDIF
Podmienka.

za IF (ELSEIF) musi ist vyraz. Je to v sulade s tym ako funguju boolovske a rozne Cmp funkcie. Neprazdne = True, Prazdne = False

 

Napr. 

 

set A "X"

set B "X"

 

if A

  prikaz1

  prikaz2

  prikaz3

elseif B

  prikaz4

  prikaz5

  prikaz6

else

  prikaz7

endif

 

pocet ELSEIF nie je obmedzeny.

pocet vnoreni nie je odmedzeny

 

Funguje to tak, ze akonahle je jeden blok splneny, tak ostatne uz sa nesplnia, aj keby boli pravdive.

cize v priklade hore, aj ke A aj B je pravda, splni sa iba to A, lebo to ide skor.

Prikazy ktore su potlacene cez IF - ELSEIF - ELSE - ENDIF sa neparsuju, t.j. pripadne chyby v nich nebudu odhalene pri kazdom spracovani.

Pri testovani teda treba pretestovat vsetky vetvy.

 

cize skript

 

set A "X"

IF A

show "aaa"

else

  blablabla 

endif

 

sa vykona, ale ak zmenis set A "X" na set A "", tak to vyhodi chybu, lebo blabla nepozna

 


>{data}
odosle {data} tak ako su zapisane, nic sa nevyhodnocuje, a za nimi CRLF
 

SEND <retazec>
odosle retazec, ktory predtym na rozdiel od > vyhodnoti, a neposiela CRLF
>A+B odosle A+B a CRLF
SEND A+B odosle hodnotu premennej A a za nou hodnotu premennej B (vyhodnoti sa A+B a odosle sa vysledok)

<{data}
zistuje, ci obsah premennej response (ktora obsahuje data nacitane zo socketu) zodpoveda {data}

{data} su pouzite tak ako su, nevyhodnocuje sa. Sposob zistovania ci data zodpovedaju obsahu response urcuje nastavenie MATCHMODE
 


RESP <retazec>
zistuje, ci obsah premennej response (ktora obsahuje data nacitane zo socketu) zodpoveda hodnote retazca

Na rozdiel od < retazec sa vyhodnocuje. Sposob zistovania ci data zodpovedaju obsahu response urcuje nastavenie MATCHMODE

!<
NOT RESP
negovane prikazy < a RESP


MATCH <premenna>, <retazec>
Podobne ako RESP, len sa automaticky nepracuje s premennou response, ale s urcenou premennou.

MATCH Response, <retazec> = RESP <retazec>. Sposob zistovania ci data zodpovedaju obsahu response urcuje nastavenie MATCHMODE

NOT MATCH
negovany prikaz MATCH


MATCHMODE <rezim>
urcuje sposob akym pracuju prikazy <, !<, RESP, NOT RESP, MATCH, NOT MATCH

<rezim> je jedna z hodnot (vyznam je popisany vedla)

sametext - podmienka je splnena, ked su porovnavane texty rovnake (nezalezi na velkosti pismen)

samestring - podmienka je splnena, ked su porovnavane stringy rovnake (zalezi na velkosti pismen)

foundtext - ak sa text (v prikaze MATCH, RESP...) nachadza niekde v danej premennej (urcenej menom, resp. Response), nezalezi na velkosti pismen

foundstring - ak sa string (v prikaze MATCH, RESP...) nachadza niekde v danej premennej (urcenej menom, resp. Response), zalezi na velkosti pismen

regexpr<MODIFIKATORY> - regularny vyraz, kde modifikatory urcuju sposob vyhodnocovania regularneho vyrazu

regexpr{+/-}IRSGMX napr. regexpr-I+RSG-MX

Ak sa neuvedie nic, tak je defaultny stav -I-R+S+G-M-X

 

Vyznam modifikatorov, a viac o regularnych vyrazoch na 

 

http://regexpstudio.com/TRegExpr/Help/RegExp_Syntax.html
 

 


SLEEP <N 0..3600 sekund>
pozastavi vykonavanie skriptu na urceny cas, ale data popri tom prijima. Vhodne pouzitie ak sa caka, ze server nieco posle, a da sa mu na to nejaky cas, za ktory by malo byt prijate to co sa ocakava.

 


TIMEOUT <N 0..3600 sekund>
Nastavenie timeoutu pre opracie pripajania, prenosu...


CONNECT <host>:<port>
vytvara spojenie s danym serverom

CONNECTSSH <host>:<port>
podobne ako CONNECT, vytvara spojenie s danym serverom, a pouziva protokol SSH

Vyzaduje nastavene premenne

SSHUser - meno pouzivatela

 

 

SSHPass alebo (SSHKeyFile a SSHKeyPass)

podla toho, ci sa ma robit autorizacia heslo, alebo verejnym klucom

 

SSHPass - heslo pre pripojenie

SSHKeyFile - meno suboru s verejnym klucom

SSHKeyPass - heslo k verejnemu klucu


CLOSE
ukoncuje aktivne spojenie

 


SET premenna <retazec>
Nastavi hodnotu retazca do urcenej premennej

 

 

PARAMETER displayparams premenna <retazec>
Podobne ako SET, nastavi hodnotu retazca do urcenej premennej

Tieto premenne ak su sifrovane, nie su sifrovane az takou silnou sifrou, viazanou na obsah skriptu - kvoli modifikacii pouzivatelom

Su zobrazovane v dialogu pre CHAT, a umoznuju jednoduche nastavenie pozadovanych parametrov pre vykonanie skriptu

Cize skript moze byt dlhy, zlozity, ale pouzivatel uvidi pri konfiguracii iba vybrane premenne - parametre

 

parameter "jano:bool" jano "X"
parameter "palo:list:111|222|333|444" palo "111222333"
parameter "miso:bool" miso %O18PJKOXED8NRVR58I6DRJN3JSN8AMAGRHAKN0A9NZ%
parameter "emil:fixedlist:AAA|BBB|CCC|DDD" emil "111222333"

 

parameter "login" login "jano"
parameter "heslo:password" pass ""

 

za nazvom (ktory sa zobrazuje) moze ist urcenie typu

 

:bool  - checkbox, bude nadobudat hodnotu "" a "T", a ked zadas inu hodnotu ako "T" pri definicii, tak sa po vypnuti a zapnuti zmeni na "T"

:list - zoznam hodnot, pricom mozes napisat aj vlastne nieco

:fixedlist - zoznam hodnot, pricom je mozne vybrat iba hodnotu zo zoznamu. Ak sa pri inicializacii zada nieco ine, tak to nie je problem, ze by skript nezbehol. Tu ide len o urcenie parametrov pre dialog.

 

:pass - ze je to heslo, t.j. hodnota bude sifrovana

 

ak je hodnota zasifrovana, tak je to automaticky :pass

v priklade to vidno na zapise miso:bool - tym ze hodnota je sifrovana, tak sa :bool ignoruje

 

parameter "heslo1" heslo1 %%

parameter "heslo2" heslo2 %.......%

parameter "heslo3:pass" heslo3 ""

parameter "heslo4:pass" heslo4 "jano"  - v dialogu budem vidiet ***********, a dialog to pri zmene nahradi sifrovanou hodnotou

 

tip, mozno zadat aj

 

parameter "aaa" a "1"

parameter "  bbb" b "1"  - t.j. odsadenie pred bbb

 

--------------------------------------

rozsirenie zobrazovania - ak sa ma parameter zobrazovat iba niekedy

 

pred parameter je potrebne dopisat 

 

{ID/podmienka pre zobrazenie}

 

pricom ak ID nie je potrebne, tak sa napise iba {/podmienka pre zobrazenie}

ak nie je potrebna podmienka, tak iba {ID}

 

riadit mozno pomocou parametrov typu

- bool 

- fixedlist 

 

cele je to riadene nasledovnou filozofiou:

 

podmienka pre zobrazenie je boolovsky vyraz,kde ako operator 

and mozno pouzit & alebo .

or mozno pouzit | alebo +

negacia je ^ za vyrazom

zatvorky

 

precedencia (priorita je), od najvyssej po najnizsiu

 

()

^

&.

|+

 

cize

 

(a.b.c)^ je nand 

 

 

premenna v podmienke zobrazenia parametra sa tyka iba parametrov, a nema to nic spolocne s chatom, skriptom,

je to defacto iba barlicka, ktora definuje kedy co zobrazit. Cize ked tam napisem xyz, tak sa nic nedeje, aj ked nic take nie je nikde definovane, existuje to ale ma to hodnotu FALSE. xyz^ ma hodnotu TRUE.

 

 

Ak ma parameter definovane <ID>, 

- tak ak je zobrazeny, premenna <ID>V ma hodnotu TRUE  (V od slova visible)

- ak nie je zobrazeny, premenna <ID>H ma hodnotu TRUE  (H od slova hiden)

 

dalej, tak je parameter bool, tak ak je zafajknuty, tak ID1, ak nie je tak ID0 ma hodnotu TRUE

ak parameter je fixedlist, tak ak je vybrata prva polozka, tak ID0, ak druha ID1, ak tretia ID2... ma hodnotu TRUE

dolezite, ID<N> je iba vtedy, ak je dany prvok zobrazeny (t.j vtedy je definovane aj IDV)

 

priklad

 

parameter "{sqlt}SQL Type:fixedlist:MS SQL|Oracle|Informix|Firebird" sqltype ""

 

// tymto som nastavil parameter - premennu chatu sqltype na prazdnu hodnotu, 

// pouzivatel to moze menit cez pevny zoznam - combobox, ktory ma 4 polozky MS SQL...Firebird

// dal som tomuto parametru pre ucely zobrazenia dialogu ID sqlt

// podmienka pre zobrazenie nie je ziadna, cize to bude vidno vzdy

// a pouzivatelovi sa zobrazi nazov SQL Type

 

parameter "login" login ""
parameter "heslo:password" pass ""

 

parameter "{/sqlt0}MS SQL Database" mssql ""

// tento parameter sa zobrazi iba ak je sqlt t.j. SQL Type nastaveny na prvu polozku (index 0) zoznamu, t.j. MS SQL


parameter "{ordb/sqlt1}Oracle Database" oracledb ""

 

// tento parameter sa zobrazi iba ak je sqlt t.j. SQL Type nastaveny na druhu polozku (index 1) zoznamu, t.j. Oracle

// a dal som tomu ID ordb


parameter "{/ordbV}Oracle A" oracleA ""

 

// tento parameter bude zobrazeny iba ak je zobrazeny Oracle Database (ma ID ordb, a ordbV je TRUE ak je ordb zobrazeny)


parameter "{/ordbV}Oracle B" oracleB ""
parameter "{/ordbV}Oracle C" oracleC ""

 

parameter "{/sqlt2}Informix Database" informixdb ""
parameter "{/sqlt3}Firebird Database" firebirddb ""

 

========================================

 

ukazka boolovskej algebry pre zobrazovanie

 

parameter "{A}A:bool" A ""
parameter "{B}B:bool" B ""
parameter "{C}C:bool" C ""

 

parameter "{/A1.B1.C1}A and B and C" X1 ""
parameter "{/A1+B1+C1}A or B or C" X2 ""
parameter "{/A1.B1+C1}A and B or C" X3 ""
parameter "{/A1+B1.C1}A or B and C" X4 ""

parameter "{/(A1+B1).C1}(A or B) and C" X5 ""

parameter "{/(A1.B1.C1)^}not (A and B and C)" X6 ""

pri chybe v zatvorkovani, ci inej syntaktickej chybe sa neobjavi ziadna hlaska, akurat parameter je vzdy zobrazeny, a to aj s definiciou podmienky..

 

 

 


SENDEMAILSMTP <smtpserver>:<port>
odosle email pouzitim SMTP protokolu

Vyzaduje nastavene premenne

EmailFrom
EmailTo
EmailSubject
EmailBody
SMTPAuthType -  sposob autorizacie (pouzitia SMTPUser, SMTPPass), moze byt jedna z hodnot : none, plain, login, crammd5, crammsha1, auto

SMTPUser
SMTPPass

CHECKEMAILPOP3 <pop3server>:<port>
prijme email poutitim POP3 protokolu

Vyzaduje nastavene premenne

CheckEmailSubject - funkcia uspeje, ak sa v schranke najde email s takymto subjektom (na velkosti pismen nezalezi). Sprava je vymazana.


DeleteEmailSubject - regularny vyraz, sluzi na preventivne ucely, aby sa v schranke nekopili testovacie emaily, a su vymazane vsetky spravy zo schranky, ktore maju subjekt vyhovujuci tomuto regularnemu vyrazu. Vid. http://regexpstudio.com/TRegExpr/Help/RegExp_Syntax.html, zapnute modifikatory su -I-R+S+G-M-X


POP3AuthType - sposob autorizacie (pouzitia POP3User, POP3Pass), moze byt jedna z hodnot : none, login, crammd5, crammsha1
POP3User
POP3Pass


SNMPVALUES <agent>
vycita z agenta urcene hodnoty

Vyzaduje nastavene premenne

SNMPComunity - napr. SET  SNMPComunity "public"

SNMPObjectID - retazec, v ktorom su bodkociarkou oddelene jednotlive OID

SNMPTimeout - napr. SET SNMPTimeout 1000 (znamena 1000ms), 1..300000

SNMPRetries - pocet pokusov o nacitanie hodnoty, 1..8

Funguje tak, ze nacita vsetky hodnoty pre zadane OID, hodnoty su v jednotlivych riadkoch vratenej premennej, t.j. prvy riadok vrateneho retazca je hodnota pre prve OID, druhy je hodnota druheho, atd...

 


SNMPTABLE <agent>
vycita z agenta tabulku hodnot

Vyzaduje nastavene premenne

SNMPComunity - napr. SET  SNMPComunity "public"

SNMPObjectID - retazec, v ktorom su bodkociarkou oddelene jednotlive OID

SNMPTimeout - napr. SET SNMPTimeout 1000 (znamena 1000ms), 1..300000

SNMPRetries - pocet pokusov o nacitanie hodnoty, 1..8

Prikaz funguje tak, ze nacita vsetky hodnoty, ktorych OID zacina OID zadanym v SNMPObjectID 

T.j. ak 

 

SET SNMPObjectID ".1.3.6.1.2.1.2.2.1.1.;.1.3.6.1.2.1.2.2.1.2.;.1.3.6.1.2.1.2.2.1.3."

(pricom ak by bolo zadane OID bez bodky na konci, tak bude bodka automaticky doplnena)

tak sa nacita kazda hodnotak, ktorej OID zacina

 

.1.3.6.1.2.1.2.2.1.1.

 

cize

 

.1.3.6.1.2.1.2.2.1.1.1

.1.3.6.1.2.1.2.2.1.1.2

.1.3.6.1.2.1.2.2.1.1.3

...

 

ale aj

 

.1.3.6.1.2.1.2.2.1.1.1.1.1

.1.3.6.1.2.1.2.2.1.1.1.2.1

.1.3.6.1.2.1.2.2.1.1.1.3.1

.1.3.6.1.2.1.2.2.1.1.2.1.1

 

pricom to co zostane sa berie ako index.

 

Kolko OID je uvedenych, tolko stlpcov sa nacita.

Nacitane hodnoty sa nachadzaju v riadkoch premennych 

 

SNMPTableCol1..SNMPTableColN

 

v premennej SNMPTableCol0 sa nachadzaju ID, ako je popisane vyssie

 

Riadky premennych SNMPTableCol navzajom koresponduju, t.j. 

 

SNMPTableCol3 ma na riadku 2 hodnotu, ktorej OID je OID cislo 3 z SNMPObjectID nasledovane hodnotou na riadku 2 premennej SNMPTableCol0

 

 

 


#     komentar, riadok co ma toto na zaciatku bude ignorovany
//     komentar, riadok co ma toto na zaciatku bude ignorovany

WAITFORRESPONSEON [<until string>]
Nastavi taky rezim spracovania, ze sa pokracuje na dalsi prikaz skriptu, az ked po odoslani dat serveru, server nieco posle nazad. T.j. caka sa na odpoved. Defaultne je tato volba automaticky zapnuta

 

Moze mat jeden volitelny parameter, 

cize napr.

waitforresponseon "sealit@salamander:~$"

Funguje to tak, ze sa stav cakania ukonci, az ked bol v nacitanych datach 
najdeny takyto retazec, pricom to kludne moze ist zo servera po 1- 2 pismenkach, (takto 
aj realne niekedy chodi).

K tomuto pripomienka dolezita, ked sa posle

SEND Path_script + " network inout" + CRLF

tak to sposobi, ze server prompt posle 2x (ako klasicka reakcia na prazdny 
enter), preto je dolezite vsetky send aby boli bud v tvare

SEND Path_script + " network inout" + LF

alebo

SEND Path_script + " network inout" + CR

inak sa to neda rozumne tymto sposobom synchronizovat.

Treba si uvedomit, ze sa chat nepouziva len na pripajanie k serverom, 
ktore vratia nejaky prompt. Http server neposle pri connecte nic.


WAITFORRESPONSEOFF
Nastavi taky rezim spracovania, ze sa odosielaju data serveru bez ohladu na to, ci nieco odpovedal alebo nie. Vhodne napr. pre http protokol, kde server po pripojeni neposle ziaden baner, ale caka na to, aku poziadavku dostane, cize bez tejto volby by chat nieco odoslal, a cakal by na nieco, co nepride. (cakal by nastaveny timeout)


SAVEVAR <premenna1> <premenna2> <premenna3> .
sposobi, ze po skonceni skriptu su obsahy tychto premennych odpamatane, a moznoich pouzivat pri specialnych funkciach pre watches...

 

Poznamka

savevar A B C D

je ekvivalentne

savevar A B
savevar C
savevar D

savevar moze byt pouzity viackrat, cize netreba davat vsetky premenne do jedneho riadku, a nezalezi ci savevar 
sa da na zaciatku alebo na konci skriptu, ono si to len nastavi flag, ze tu premennu treba si odpamatat, ked skript skonci.


SHOW <retazec>
sluzi pre ucely ladenia, zobrazi hodnotu retazca, ak retazec nie je uvedeny (t.j. iba samotny SHOW prikaz) sposobi vypis prazdneho riadku

 

ExecWSHScript skript premenna1 premenna2 ... premennaN
sluzi na spustenie skriptu - vbs, java... (standardne dostupne)

 

vyzaduje definovane premenne s nasledovnymi hodnotami napr.

 

WSHScriptEngine "vbscript"
WSHScriptTimeoutSec "10"
WSHScriptMaxMemoryMB "8"

Preroze script engine sa moze zblaznit, ci moze byt zle napisany skript, tak cmonitor sleduje parametre tohto procesu, a ked to bezi dlho, alebo to zerie vela pamati, tak je to nasilne ukoncene (kilnute).

 

Odchytava sa ako STDOUT tak STDERR, cize ked je v skripte nejaka chyba, tak 
to napise do stdout, a to potom ja aj zobrazim ako dovod zlyhania funkcie.

Odporucam tak ako je adresar Modules, vytvorit aj adresar ci uz scripts, 
alebo wshscripts. Aby v tom bol poriadok.

 

set A "1"
set C "3"

set WSHScriptEngine "vbscript"
set WSHScriptTimeoutSec "10"
set WSHScriptMaxMemoryMB "8"

ExecWSHScript ".\Script\sample1.vbs" A "BCDEFGH" C

show aa
show bb
show cc

============================

rozhranie, ukazka vo vbs scripte pre odovzdavanie a prijem parametrov chatu:

 

Set WshShell = WScript.CreateObject("WScript.Shell")
Set WshProcEnv = WshShell.Environment("PROCESS")

 

'WshProcEnv sa pouzije pre pristup k hodnote - parametru z chatu
'parametre sa prenasaju cez environment procesu, pod menom
'CHATPARAM_<poradove cislo pocnuc 1>

 


A = WshProcEnv("CHATPARAM_1")
B = WshProcEnv("CHATPARAM_2")
C = WshProcEnv("CHATPARAM_3")

 


'Nasleduje samotne telo skriptu, ktore ma robit nieco uzitocne
'STDERR je tiez presmerovany a vyhodnoteny samostatne
'a tato chyba je potom publikovana ako dovod zlyhania skriptu

 

AA = A & "zmena"
BB = AA & B
CC = BB & C

 

'takto napr. sa uspi skript na 1000ms, ale je to teraz odpoznamkovane
'WScript.Sleep 1000

 

'Vystup sa deje presmerovanim STDOUT, preto treba zapisovat na STDOUT
'Chat vystup parsuje, a ak najde CHATVAR_<premenna>=hodnota#0
'tak nastavi danu premennu na urcenu hodnotu - v tomto pripade
'sa nastavia premenne AA,BB a CC
'pred vystup premennych je dobre dat prazdny riadok, 
'CHAT vystup parsuje, hlada CHATVAR_ pred ktorym je nic alebo biely znak
'a to coje medzi CHATVAR_ a = berie ako meno premennej, max dlzka je 50znakov
'velkost hodnoty nie je nejako obmedzena, iba programovo su odignorovane vystupy 
'vacsie ako 64k - nie je predpoklad ze by chat mal spracovavat daco az takto velke

 

WScript.StdOut.WriteLine
WScript.StdOut.WriteLine "CHATVAR_AA=" & AA &chr(0)
WScript.StdOut.WriteLine "CHATVAR_BB=" & BB &chr(0)
WScript.StdOut.WriteLine "CHATVAR_CC=" & CC &chr(0)

 

 

 

 

 

 

 

Premenne, konstanty
====================================
GetUniqueString
vrati jedinecny retazec, je tvoreny (pre verziu 2.0.1.285 a nizsie) 

RRRRRRRRRYYMMDDHHNNSSZZZTTTTTTTT

 

RRRRRRRRR je 9 nahodnych velkych pismen A-Z

YY rok (posledne dvojcislie)  00-99

MM mesiac 01-12

DD den 01-31

HH hodina 00-23

NN minuta 00-59

SS sekunda 00-59

ZZZ milisekunda 000-999

TTTTTTTT pocet ms od spustenia OS hexadecimalne (00000000-FFFFFFFF)

 

Vhodne na generovanie unikatnych subjektov pre napr. loop test odoslat cez SMTP a prijat cez POP3

 

CRLF - odriadkovanie 0D0A

 

CR - 0D

 

LF - 0A

 

TAB - 09

 

Timeout - nastaveny timeout v sekundach, nastavuje sa prikazom Timeout N

 

Response - neda sa nastavit, obsahuje nacitane data (co vratil server), je mazany niektorymi prikazmi

 

MatchSubExpr0..MatchSubExpr255 - obsahuje retazce ziskane pri porovnavani pomocou regularnych vyrazov (v prikazoch RESP, MATCH...)

 

SNMPTableCol0..SNMPTableCol255 - obsahuje nacitane stlpce prikazom SMTPTable, SNMPTableCol0 obsahuje ID riadku (tvorene OID danej hodnoty, pricom zo zaciatku je odstranene OID uvedene v premennej SNMPObjectID), a SNMPTableCol1..SNMPTableCol255 obsahuju hodnoty pre jednotlive OID

 

===================================================

 

Modifikátor @
 

modifikator @ funguje tak, ze

 

SNMPTable 123.123.123.123

 

je mozne zapisat bud tak ako to je (kvoli spatnej kompatibilite)

 

alebo

 

SNMPTable @"123.123.123.123"

 

alebo

 

SET IP "123.123.123.123"

SNMPTable @IP

 

alebo

 

SNMPTable @Set( IP, "123.123.123.123" )

Show IP

 

Cize modifikator @ znamena, ze sa to ma vyhodnotit ako vyraz, to co je vyssie su uz len dosledky z toho plynuce. 

Tento modifikator je mozne pouzit vo funkciach :

 

SLEEP

TIMEOUT

CONNECT

CONNECTSSH

SENDEMAILSMTP

CHECKEMAILPOP3

SNMPVALUES

SNMPTABLE

 

 

